

Unraveling Adaptive Evolutionary Divergence at Microgeographic Scales

Supporting File for Clancey et al.

Here we derive the equations in the appendix corresponding to the analytical model for phenotypic change. Our analytical model follows the life cycle presented in Figure 2 in the main text.

Throughout we will use $z[i]$ to denote the phenotype of individuals within habitat i and $Z[i]$ to denote the average within population i . Below we begin derive the recursion equation for the the mean phenotype in each habitat, solve for the mean phenotype at equilibrium and express the divergence between the mean phenotype in the two habitats.

Assumptions: We want to assume that selection is weak and that habitat selection is weak. To ensure that selection is weak (even on migrants) we have to assume that an individual's phenotype, z , is close to the phenotypic optimum in both population 1 AND in population 2. This means that $(z - \theta[1]) = O(\epsilon)$ and $(z - \theta[2]) = O(\epsilon)$. To ensure that both of these are true, we let $\theta = \frac{\theta[2] + \theta[1]}{2}$, the mean of the optima. We then can insure that selection is always weak (not just within a single population but between populations) by letting that the deviation of an individual's phenotype from the average optima be $\delta z \in (z - \theta)$ and that the distance between the two optima be $\delta\theta \in \theta[2] - \theta[1]$ where ϵ is small. We use the following substitutions throughout:

$$\begin{aligned} \text{In } [\epsilon] := \quad & \mathbf{zSub} = \left\{ z \rightarrow \theta + \delta z \epsilon, \theta[1] \rightarrow \theta - \frac{\delta\theta}{2} \epsilon, \theta[2] \rightarrow \theta + \frac{\delta\theta}{2} \epsilon, \right. \\ & \left. Z[1, t] \rightarrow \theta + \epsilon \delta Z[1, t], Z[2, t] \rightarrow \theta + \epsilon \delta Z[2, t], Vz \rightarrow Vz \epsilon^2 \right\}; \\ \text{In } [\epsilon] := \quad & \mathbf{zSubRev} = \left\{ \delta z \rightarrow \frac{z - \frac{\theta[2] + \theta[1]}{2}}{\epsilon}, \delta Z[1, t] \rightarrow \frac{Z[1, t] - \frac{\theta[2] + \theta[1]}{2}}{\epsilon}, \right. \\ & \left. \delta Z[2, t] \rightarrow \frac{Z[2, t] - \frac{\theta[2] + \theta[1]}{2}}{\epsilon}, \delta\theta \rightarrow \frac{\theta[2] - \theta[1]}{\epsilon}, \theta \rightarrow \frac{\theta[2] + \theta[1]}{2}, Vz \rightarrow \frac{Vz}{\epsilon^2} \right\}; \end{aligned}$$

Mating and Reproduction

The probability a female with phenotype z_F mates with a male with phenotype z_M is given by:

$$\text{In } [\epsilon] := \quad \mathbf{Pmate}[z_F, z_M] := \text{Exp}[-\alpha (z_M - z_F - \delta)^2]$$

In the analytical model we assume that there is no assortative mating $\alpha=0$.

In[*]:= **Pmate**[zF, zM] /. $\alpha \rightarrow 0$
 Out[*]:= 1

As a result the mean phenotype does not change after mating and reproduction. The mean phenotype after reproduction (ZR) is therefore equal to the mean phenotype before reproduction (Z).

In[*]:= **ZR**[i_, t_] := **Z**[i, t]

Phenotype-dependent Dispersal

The probability that an individual with phenotype z migrates from habitat j to habitat i is given by:

In[*]:= **Pmig**[i_, j_, z_] := 2 m H[i] $\frac{\text{Exp}[-\eta (z - \theta[i])^2]}{\text{Sum}[\text{Exp}[-\eta (z - \theta[k])^2], \{k, 1, 2\}]}$

The mean phenotype in population i after migration (ZM) is given by the following integral where $fzR[i, z]$ is the probability density of individuals with phenotype z in population i after reproduction.

ZM[i_, t_] =

$$\frac{1}{C} \int_{-\infty}^{\infty} z (fzR[i, z] \times N[i] (1 - Pmig[j, i, z]) + fzR[j, z] \times N[j] \times Pmig[i, j, z]) dz$$

The argument in parentheses is the relative probability density of individuals of phenotype z in population i after migration. Here C is a normalization constant that normalizes this distribution to have an area of 1.

We can rewrite the integral as the sum of three integrals

$$\begin{aligned} \text{ZM}[1, t] = & \int_{-\infty}^{\infty} (z fR[1, z] \times NH[1]) dz + \int_{-\infty}^{\infty} (-z fR[1, z] \times NH[1] \times Pmig[2, 1, z]) dz + \\ & \int_{-\infty}^{\infty} (z fR[2, z] NH[2] \times Pmig[1, 2, z]) dz \end{aligned}$$

By integrating the distribution $fzR[i]$ over z gives an expectation: $\int_{-\infty}^{\infty} fzR[i, z] A dz = E_{z1i}[A]$. We can perform such an expectation in *Mathematica* by substituting z^2 for $ZR[i, t]^2 + V_z$ where V_z is the phenotypic variance in habitat i (note we assume the phenotypic variance is equal in both habitats) and z for $ZR[i, t]$. Note that like the genetic variance we will assume the the phenotypic variance remains constant throughout.

The Un-normalized Change

We are going to work out the integrals for the change in the phenotype in population 1. We will then generalize the result to an arbitrary population

```

In[*]:= int1 = Collect[Normal[Series[z NH[1] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[1, t]^2 + Vz, z → ZR[1, t]} // Factor

Out[*]=
      NH[1] × Z[1, t]

In[*]:= int2 =
      Collect[Normal[Series[-NH[1] z Pmig[2, 1, z] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[1, t]^2 + Vz, z → ZR[1, t]} // Factor // Simplify

Out[*]=
      1
      - m NH[1] × H[2] (-η (θ[1] - θ[2]) (θ[1] + θ[2])^2 + 2 Z[1, t] (-2 + η (θ[1]^2 - θ[2]^2)))
      4

In[*]:= int3 = Collect[Normal[Series[NH[2] z Pmig[1, 2, z] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[2, t]^2 + Vz, z → ZR[2, t]} // Factor // Simplify

Out[*]=
      1
      - m NH[1] × H[2] (-η (θ[1] - θ[2]) (θ[1] + θ[2])^2 + 2 Z[2, t] (2 + η (θ[1]^2 - θ[2]^2)))
      4

```

Normalizing Constant

The normalization constant is

$C = \int_{-\infty}^{\infty} f_i[i, z] \times N[i] (1 - Pmig[j, i, z]) + f_j[j, z] \times N[j] \times Pmig[i, j, z] dz$. As with the integral above we can rewrite this as the sum of three integrals.

$$C = \int_{-\infty}^{\infty} (f_1[1, z] NH[1]) dz + \int_{-\infty}^{\infty} (-f_1[1, z] NH[1] \times Pmig[2, 1, z]) dz + \int_{-\infty}^{\infty} (f_1[2, z] NH[2] \times Pmig[1, 2, z]) dz$$

```

In[*]:= intC1 = Collect[Normal[Series[NH[1] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[1, t]^2 + Vz, z → ZR[1, t]} // Factor

Out[*]=
      NH[1]

In[*]:= intC2 =
      Collect[Normal[Series[-NH[1] × Pmig[2, 1, z] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[1, t]^2 + Vz, z → ZR[1, t]} // Factor

Out[*]=
      1
      - m NH[1] × H[2] (-2 + 2 η Z[1, t] × θ[1] - η θ[1]^2 - 2 η Z[1, t] × θ[2] + η θ[2]^2)
      2

In[*]:= intC3 =
      Collect[Normal[Series[NH[2] × Pmig[1, 2, z] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[2, t]^2 + Vz, z → ZR[2, t]} // Factor

Out[*]=
      1
      - m NH[1] × H[2] (2 + 2 η Z[2, t] × θ[1] - η θ[1]^2 - 2 η Z[2, t] × θ[2] + η θ[2]^2)
      2

```

The mean phenotype after dispersal

Generalizing to an arbitrary population

```

In[*]:= ZM[i_, t_] := Normal[Series[Factor[ $\frac{\text{int1} + \text{int2} + \text{int3}}{\text{intC1} + \text{intC2} + \text{intC3}}$ ] /. zSub, {ϵ, 0, 2}]] /. zSubRev /.
      {Z[1, t] → Z[i, t], Z[2, t] → Z[Mod[i, 2] + 1, t], H[1] → H[i],
      H[2] → H[Mod[i, 2] + 1], θ[1] → θ[i], θ[2] → θ[Mod[i, 2] + 1]} // Simplify

In[*]:= ZM[1, t] /. zSubRev // Simplify
Out[*]= (1 - m H[2]) Z[1, t] + m H[2] × Z[2, t]

In[*]:= ZM[2, t] /. zSubRev // Simplify
Out[*]= m H[1] × Z[1, t] + (1 - m H[1]) Z[2, t]

```

Natural Selection

The change in the mean phenotype due to selection is given by Lande 1976 Equation 7. $\Delta Z = G \frac{1}{\bar{W}} \frac{d}{dz} \bar{W}$ where \bar{W} is the average fitness

```

In[*]:= W[z_, i_] := Exp[-γ (z - θ[i])2]

Let Y[i,t] be the mean phenotype after migration. We are going to use this as a stand-in until we want to
plug in the full expression from above.

In[*]:= Wavg[i_] := Expand[Normal[Series[W[z, i] /. zSub, {ϵ, 0, 2}]] /. zSubRev /. Z → Y[i, t]] /.
      {z2 → Y[i, t]2 + Vz, z → Y[i, t]} // FullSimplify

In[*]:= G  $\left( \frac{D[Wavg[1], Y[1, t]]}{Wavg[1]} \right)$ 
Out[*]=  $-\frac{2 G \gamma (Y[1, t] - \theta[1])}{1 - Vz \gamma - \gamma (Y[1, t] - \theta[1])^2}$ 

```

The phenotype after selection is given by:

```

In[*]:= ZS[i_, t_] := Normal[Series[Y[i, t] + G  $\left( \frac{D[Wavg[i], Y[i, t]]}{Wavg[i]} \right)$  /.
      {Y[1, t] → ZM[1, t], Y[2, t] → ZM[2, t]} /. zSub, {ϵ, 0, 2}]] /. zSubRev // Simplify

In[*]:= ZS[1, t]
Out[*]= (-1 + 2 G γ) (-1 + m H[2]) Z[1, t] + m (1 - 2 G γ) H[2] × Z[2, t] + 2 G γ θ[1]

```

```
In[*]:= ZS[2, t]
Out[*]= m (1 - 2 G γ) H[1] × Z[1, t] + (-1 + 2 G γ) (-1 + m H[1]) Z[2, t] + 2 G γ θ[2]
```

Equilibrium

```
In[*]:= equi = Solve[{(ZS[1, t]) == Z[1, t], ZS[2, t] == Z[2, t]}, {Z[1, t], Z[2, t]}] // Simplify
Out[*]=
```

$$\left\{ \left\{ \begin{aligned} Z[1, t] &\rightarrow \frac{-m (H[1] \times \theta[1] + H[2] \times \theta[2]) + 2 G \gamma ((-1 + m H[1]) \theta[1] + m H[2] \times \theta[2])}{-m (H[1] + H[2]) + 2 G \gamma (-1 + m (H[1] + H[2]))} , \\ Z[2, t] &\rightarrow \frac{-2 G \gamma \theta[2] + m (-1 + 2 G \gamma) (H[1] \times \theta[1] + H[2] \times \theta[2])}{-m (H[1] + H[2]) + 2 G \gamma (-1 + m (H[1] + H[2]))} \end{aligned} \right\} \right\}$$

```
In[*]:= Z[2, t] - Z[1, t] /. equi[[1]] /. zSub /. zSubRev /. H[1] → 1 - H[2] // Simplify
Out[*]=
```

$$\frac{2 G \gamma (\theta[1] - \theta[2])}{-2 G \gamma + m (-1 + 2 G \gamma)}$$