

# Unraveling Adaptive Evolutionary Divergence at Microgeographic Scales

Supporting File for Clancey et al.

Here we derive the equations in the appendix corresponding to the analytical model for phenotypic change. Our analytical model follows the life cycle presented in Figure 1 in the main text.

Throughout we will use  $z[i]$  to denote the phenotype of individuals within habitat  $i$  and  $Z[i]$  to denote the average within population  $i$ . Below we begin derive the recursion equation for the the mean phenotype in each habitat, solve for the mean phenotype at equilibrium and express the divergence between the mean phenotype in the two habitats.

**Assumptions:** We want to assume that selection is weak and that habitat selection is weak. To ensure that selection is weak (even on migrants) we have to assume that an individual's phenotype,  $z$ , is close to the phenotypic optimum in both population 1 AND in population 2. This means that  $(z - \theta[1]) = O(\epsilon)$  and  $(z - \theta[2]) = O(\epsilon)$ . To ensure that both of these are true, we let  $\theta = \frac{\theta[2] + \theta[1]}{2}$ , the mean of the optima. We then can insure that selection is always weak (not just within a single population but between populations) by letting that the deviation of an individuals phenotype from the average optima be  $\delta z \in (z - \theta)$  and that the distance between the two optima be  $\delta\theta \in \theta[2] - \theta[1]$  where  $\epsilon$  is small. We use the following substitutions throughout:

$$\begin{aligned} \text{In[1]:= } \mathbf{zSub} &= \left\{ z \rightarrow \theta + \delta z \epsilon, \theta[1] \rightarrow \theta - \frac{\delta\theta}{2} \epsilon, \right. \\ &\quad \left. \theta[2] \rightarrow \theta + \frac{\delta\theta}{2} \epsilon, Z[1, t] \rightarrow \theta + \epsilon \delta Z[1, t], Z[2, t] \rightarrow \theta + \epsilon \delta Z[2, t] \right\}; \\ \text{In[2]:= } \mathbf{zSubRev} &= \left\{ \delta z \rightarrow \frac{z - \frac{\theta[2] + \theta[1]}{2}}{\epsilon}, \delta Z[1, t] \rightarrow \frac{Z[1, t] - \frac{\theta[2] + \theta[1]}{2}}{\epsilon}, \right. \\ &\quad \left. \delta Z[2, t] \rightarrow \frac{Z[2, t] - \frac{\theta[2] + \theta[1]}{2}}{\epsilon}, \delta\theta \rightarrow \frac{\theta[2] - \theta[1]}{\epsilon}, \theta \rightarrow \frac{\theta[2] + \theta[1]}{2} \right\}; \end{aligned}$$

---

## Mating and Reproduction

The probability a female with phenotype  $z_F$  mates with a male with phenotype  $z_M$  is given by:

$$\text{In[3]:= } \mathbf{Pmate}[zF\_ , zM\_ ] := \text{Exp}[-\alpha (zM - zF - \delta)^2]$$

In the analytical model we assume that there is no assortative mating  $\alpha=0$ .

```
In[4]:= Pmate[zF, zM] /. α → 0
```

```
Out[4]= 1
```

As a result the mean phenotype does not change after mating and reproduction. The mean phenotype after reproduction (ZR) is therefore equal to the mean phenotype before reproduction (Z).

```
In[5]:= ZR[i_, t_] := Z[i, t]
```

## Phenotype-dependent Dispersal

The probability that an individual with phenotype  $z$  migrates from habitat  $j$  to habitat  $i$  is given by:

$$\text{In[6]:= Pmig}[i_, j_, z_] := 2 m H[i] \frac{\text{Exp}[-\eta (z - \theta[i])^2]}{\text{Sum}[\text{Exp}[-\eta (z - \theta[k])^2], \{k, 1, 2\}]}$$

The mean phenotype in population  $i$  after migration (ZM) is given by the following integral where  $fzR[i, z]$  is the probability density of individuals with phenotype  $z$  in population  $i$  after reproduction.

$$ZM[i_, t_] =$$

$$\frac{1}{C} \int_{-\infty}^{\infty} z (fzR[i, z] \times N[i] (1 - Pmig[j, i, z]) + fzR[j, z] \times N[j] \times Pmig[i, j, z]) dz$$

The argument in parentheses is the relative probability density of individuals of phenotype  $z$  in population  $i$  after migration. Here  $C$  is a normalization constant that normalizes this distribution to have an area of 1.

We can rewrite the integral as the sum of three integrals

$$ZM[1, t] = \int_{-\infty}^{\infty} (z fR[1, z] \times NH[1]) dz + \int_{-\infty}^{\infty} (-z fR[1, z] \times NH[1] \times Pmig[2, 1, z]) dz + \int_{-\infty}^{\infty} (z fR[2, z] NH[2] \times Pmig[1, 2, z]) dz$$

By integrating the distribution  $fzR[i]$  over  $z$  gives an expectation:  $\int_{-\infty}^{\infty} fzR[i, z] A dz = E_{z1,i}[A]$ . We can perform such an expectation in *Mathematica* by substituting  $z^2$  for  $ZR[i, t]^2 + V_z$  where  $V_z$  is the phenotypic variance in habitat  $i$  (note we assume the phenotypic variance is equal in both habitats) and  $z$  for  $ZR[i, t]$ . Note that like the genetic variance we will assume the the phenotypic variance remains constant throughout.

## The Un-normalized Change

We are going to work out the integrals for the change in the phenotype in population 1. We will then generalize the result to an arbitrary population

```

In[7]:= int1 = Collect[Normal[Series[z NH[1] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[1, t]^2 + Vz, z → ZR[1, t]} // Factor
Out[7]= NH[1] × Z[1, t]

In[8]:= int2 =
      Collect[Normal[Series[-NH[1] z Pmig[2, 1, z] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[1, t]^2 + Vz, z → ZR[1, t]} // Factor // Simplify
Out[8]=  $\frac{1}{4} m NH[1] \times H[2] \left( -\eta (\theta[1] - \theta[2]) (\theta[1] + \theta[2])^2 + 2 Z[1, t] (-2 + \eta (\theta[1]^2 - \theta[2]^2)) \right)$ 

In[9]:= int3 = Collect[Normal[Series[NH[2] z Pmig[1, 2, z] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[2, t]^2 + Vz, z → ZR[2, t]} // Factor // Simplify
Out[9]=  $\frac{1}{4} m NH[1] \times H[2] \left( -\eta (\theta[1] - \theta[2]) (\theta[1] + \theta[2])^2 + 2 Z[2, t] (2 + \eta (\theta[1]^2 - \theta[2]^2)) \right)$ 

```

## Normalizing Constant

The normalization constant is

$C = \int_{-\infty}^{\infty} f_i[i, z] \times N[i] (1 - Pmig[j, i, z]) + f_j[j, z] \times N[j] \times Pmig[i, j, z] dz$ . As with the integral above we can rewrite this as the sum of three integrals.

$$C = \int_{-\infty}^{\infty} (f1[1, z] NH[1]) dz + \int_{-\infty}^{\infty} (-f1[1, z] NH[1] \times Pmig[2, 1, z]) dz + \int_{-\infty}^{\infty} (f1[2, z] NH[2] \times Pmig[1, 2, z]) dz$$

```

In[10]:= intC1 = Collect[Normal[Series[NH[1] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[1, t]^2 + Vz, z → ZR[1, t]} // Factor
Out[10]= NH[1]

In[11]:= intC2 =
      Collect[Normal[Series[-NH[1] × Pmig[2, 1, z] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[1, t]^2 + Vz, z → ZR[1, t]} // Factor
Out[11]=  $\frac{1}{2} m NH[1] \times H[2] \left( -2 + 2 \eta Z[1, t] \times \theta[1] - \eta \theta[1]^2 - 2 \eta Z[1, t] \times \theta[2] + \eta \theta[2]^2 \right)$ 

In[12]:= intC3 =
      Collect[Normal[Series[NH[2] × Pmig[1, 2, z] /. zSub, {ϵ, 0, 2}]] /. zSubRev, {z, m}] /.
      {z^2 → ZR[2, t]^2 + Vz, z → ZR[2, t]} // Factor
Out[12]=  $\frac{1}{2} m NH[1] \times H[2] \left( 2 + 2 \eta Z[2, t] \times \theta[1] - \eta \theta[1]^2 - 2 \eta Z[2, t] \times \theta[2] + \eta \theta[2]^2 \right)$ 

```

## The mean phenotype after dispersal

Generalizing to an arbitrary population

```
In[13]:= ZM[i_, t_] := Normal[Series[Factor[ $\frac{\text{int1} + \text{int2} + \text{int3}}{\text{intC1} + \text{intC2} + \text{intC3}}$ ] /. zSub, {ϵ, 0, 2}]] /. zSubRev /.
  {Z[1, t] → Z[i, t], Z[2, t] → Z[Mod[i, 2] + 1, t], H[1] → H[i],
   H[2] → H[Mod[i, 2] + 1], θ[1] → θ[i], θ[2] → θ[Mod[i, 2] + 1]} // Simplify

In[14]:= ZM[1, t] /. zSubRev // Simplify

Out[14]= (1 - m H[2]) Z[1, t] + m H[2] × Z[2, t]
```

## Selection

The change in the mean phenotype due to selection is given by Lande 1976 Equation 7.  $\Delta Z = G \frac{1}{\bar{W}} \frac{d}{dz} \bar{W}$  where  $\bar{W}$  is the average fitness

```
In[15]:= W[z_, i_] := Exp[-γ (z - θ[i])2]

Let Y[i,t] be the mean phenotype after migration. We are going to use this as a stand-in until we want to
plug in the full expression from above.
```

```
In[16]:= Wavg[i_] := Expand[Normal[Series[W[z, i] /. zSub, {ϵ, 0, 2}]] /. zSubRev /. Z → Y[i, t]] /.
  {z2 → Y[i, t]2 + Vz, z → Y[i, t]} // FullSimplify
```

```
In[17]:= G  $\left( \frac{D[Wavg[1], Y[1, t]]}{Wavg[1]} \right)$ 
```

```
Out[17]=  $-\frac{2 G \gamma (Y[1, t] - \theta[1])}{1 - V z \gamma - \gamma (Y[1, t] - \theta[1])^2}$ 
```

The phenotype after selection is given by:

```
In[18]:= ZS[i_, t_] := Normal[Series[Y[i, t] + G  $\left( \frac{D[Wavg[i], Y[i, t]]}{Wavg[i]} \right)$  /.
  {Y[1, t] → ZM[1, t], Y[2, t] → ZM[2, t]} /. zSub, {ϵ, 0, 2}]] /. zSubRev // Simplify
```

```
In[19]:= ZS[1, t]
```

```
Out[19]=  $\frac{-((-1 + 2 G \gamma + V z \gamma) \times (-1 + m H[2]) Z[1, t]) + m (-1 + 2 G \gamma + V z \gamma) H[2] \times Z[2, t] - 2 G \gamma \theta[1]}{-1 + V z \gamma}$ 
```

## Equilibrium

In[20]:= **equi = Solve[{(ZS[1, t]) == Z[1, t], ZS[2, t] == Z[2, t]}, {Z[1, t], Z[2, t]}] // Simplify**

$$\text{Out[20]= } \left\{ \left\{ \begin{aligned} Z[1, t] &\rightarrow \frac{m(-1 + Vz \gamma) (H[1] \times \theta[1] + H[2] \times \theta[2]) + 2 G \gamma ((-1 + m H[1]) \theta[1] + m H[2] \times \theta[2])}{m(-1 + Vz \gamma) (H[1] + H[2]) + 2 G \gamma (-1 + m (H[1] + H[2]))} \end{aligned} \right. \right. \\ \left. \left. Z[2, t] \rightarrow \frac{-2 G \gamma \theta[2] + m(-1 + 2 G \gamma + Vz \gamma) (H[1] \times \theta[1] + H[2] \times \theta[2])}{m(-1 + Vz \gamma) (H[1] + H[2]) + 2 G \gamma (-1 + m (H[1] + H[2]))} \right\} \right\}$$

In[21]:= **Z[2, t] - Z[1, t] /. equi[[1]] /. zSub // Simplify**

$$\text{Out[21]= } - \frac{2 G \gamma \delta \theta \epsilon}{m(-1 + Vz \gamma) (H[1] + H[2]) + 2 G \gamma (-1 + m (H[1] + H[2]))}$$